
// ============================================================
// SENDIT.WCM SendIT! macro for WordPerfect 5.1 (Windows)
// Sends the current document as an attached file.
// ============================================================
//
APPLICATION(WP;WPWP;Default;"WPWPUS.WCD")
CALL(Init@)
CALL(GetDoc@)
CALL(DLLLoad@)
CALL(DLLCall@)
CALL(DLLFree@)
QUIT

LABEL(Init@)

// ===========================================================
// Initialize our variables ...
// ============================================================

To:=""
CC:=""
BCC:=""
Subject:=""
Attach:=""
InteractFlag:=True
NotifyFlag:=True
Dummy1:=""
Dummy2:=""
Dummy3:=""

RETURN

LABEL(GetDoc@)
// ===========================================================
// Figure out if we've got selected text for use in the message
// Otherwise we'll use stock text and attach the file.
// ===========================================================

GetWPData
(

MacroVariable:SMA;
SystemVariable:SelectModeActive!

)
If(SMA=False)

MsgText:="See Attached File ..."
    FileSave()

GetWPData
(

MacroVariable:Path;



SystemVariable:path!
)
GetWPData
(

MacroVariable:Name;
SystemVariable:Name!

)
Attach:=Path+Name
Subject:="Attached file: "+Name

Else
GetWPData
(

MacroVariable:MSGText;
SystemVariable:SelectedText!

)
EndIf

RETURN



Label(DLLLoad@)
// ============================================================
// Set up the alias to SENDIT so we can call it ...
// ============================================================

DllLoad (Mail;"SENDIT.DLL")
RETURN

LABEL(DllCall@)
// ============================================================
// Make the actual call to MAIL (SendIT.DLL) function 
// SendMHSMESSAGE, and pass it our info ...
// ============================================================

DllCall (Mail;"SendMHSMessage";Result:INTEGER;
{To;
 CC;
 BCC;
 Subject;
 MsgText;
 Attach;
 InteractFlag;
 NotifyFlag;
 Dummy1;
 Dummy2;
 Dummy3
}
)

RETURN

LABEL(DLLFree@)
// ============================================================
// Free up our DLL resources before we blow ...
// ============================================================

DllFree(Mail)
RETURN


